
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.2.2.3c which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

2 Table of Contents

11
Overview

2
Table of Contents
1
3
Verification Test Summary
2
4
Corrections required for test case 12.2.2.3c
3
4.1
Introduction
3
4.2
TC_12_2_2_3c()
4
4.3
f_TC_12_2_2_3c_UTRAN()
5
4.4
fl_TC_12_2_2_3c_Body()
5
4.5
cs_GMM_Cause
13
5
Branches executed in test case 12.2.2.3c
13
6
Execution Log Files
13
6.1
Intel 7360 device
13
7
References
14


3 Verification Test Summary

Test Case:
TC_12_2_2_3c
Test Group:
/12/UTRAN34_GMM_SIMTC.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D16wk24 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7360
Verification Status:
PASS
4 Corrections required for test case 12.2.2.3c
4.1 Introduction

This section describes the changes required to make test case 12.2.2.3c run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D16wk24 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 TC_12_2_2_3c() 
	Record Name
	TC_12_2_2_3c()

	Reason for change
	In the current TTCN implementation the guard timer for this testcase is set to 300s. Considering that there is a T3302 timer of 3 minutes in this testcase, the guardtimer value of 300s is insufficient. This needs to be increased. 

	Summary of change
	Increased the guard timer to 420s. Please see screenshot for details.

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:
	testcase TC_12_2_2_3c() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Combined PS attach / congestion / GPRS services only
    var UTRAN_IRAT_PTC   v_UTRAN      := null;
    var GERAN_PTC        v_GERAN      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(300);  
    v_UTRAN := UTRAN_IRAT_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(system, v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_2_2_3c_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop_UTRAN(t_GuardTimer);


After:

	testcase TC_12_2_2_3c() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Combined PS attach / congestion / GPRS services only
    var UTRAN_IRAT_PTC   v_UTRAN      := null;
    var GERAN_PTC        v_GERAN      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(420);//WA#12_2_2_3c 300->420
    v_UTRAN := UTRAN_IRAT_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(system, v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_2_2_3c_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop_UTRAN(t_GuardTimer);


4.3 f_TC_12_2_2_3c_UTRAN()
	Record Name
	f_TC_12_2_2_3c_UTRAN()

	Reason for change
	In the current TTCN implementation of this function, the UE is not switched OFF after the preamble (Call to function f_UTRAN34_Preamble). As per the Step 2 of the expected test sequence in 34.123-1 v 12.4.0 Sec 12.2.2.3c.4, the UE should be Switched ON or powered up. This is not possible until the UE is in a switched Off state.

	Summary of change
	The UE needs to be switched OFF after the preamble. Called the function f_UTRAN34_SwitchPower_Off after the call to function f_UTRAN34_Preamble.
Please see screenshot for details.

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	<<SKIPPED CODE>>

f_UTRAN34_SendDefSysInfo(utran34_CellA);
    //The UE has a valid TMSI-1, P-TMSI-1 and RAI-1. MS is Idle Updated on cell A
    f_UTRAN34_Preamble(utran34_CellA);
    //ATT flag is set to FALSE
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellA, tsc_AttOff);
<<SKIPPED CODE>>


After:

	<<SKIPPED CODE>>

f_UTRAN34_SendDefSysInfo(utran34_CellA);
    //The UE has a valid TMSI-1, P-TMSI-1 and RAI-1. MS is Idle Updated on cell A
    f_UTRAN34_Preamble(utran34_CellA);
    f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE);//WA#12_2_2_3c
    //ATT flag is set to FALSE
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellA, tsc_AttOff);
<<SKIPPED CODE>>


4.4 fl_TC_12_2_2_3c_Body()
	Record Name
	Fl_TC_12_2_2_3c_Body()

	Reason for change
	1) As per Step 4 of the expected test sequence in 34.123-1 v 12.4.0, Sec 12.2.2.3c.4, the UE is expected to perform an ATTACH REQ with mobile identity set as IMSI, and the TMSI Status set to “No valid TMSI available”. 

However, in the preamble, the UE has been assigned PTMSI-1, RAI-1 and TMSI as well. Hence, the UE will include the PTMSI as the mobile identity, and not IMSI. So the ATTACH REQ should expect a PTMSI, instead of IMSI. This needs to be changed in TTCN, and a Prose CR will be raised for this.

Also, as per 24.008 Sec 9.4.1.3, the UE will include TMSI Status if the UE performs a Combined GPRS/IMSI attach, and NO TMSI is available. However, a TMSI is assigned in the preamble. So, the UE will not include the IE TMSI Status in the ATTACH REQ at Step 4. 

The current TTCN expects the TMSI Status, and needs to be changed. A Prose CR will be raised for this.
2) As per Step 12 of the expected test sequence in 34.123-1v12.4.0 Sec 12.2.2.3c.4, the UE should transit an ATTACH REQ, and the subsequent test steps complete the ATTACH procedure. However, at Step 8 of the expected test sequence, the UE has been sent an ATTACH ACCEPT with update result as “GPRS only attached”, GMM cause as “congestion (cause #22)” and the timer T3302 that is set to 3 min.
As per 24.008 Sec 4.7.3.2.3.2, when the UE receives an ATTACH ACCEPT with cause #22, the UE shall change the state to GMM-REGISTERED.ATTEMPTING-TO-UPDATE-MM, shall stop the timer T3310 if running and shall enter state MM-IDLE. The UE shall set the ROUTING AREA UPDATING attempt counter to 5, and shall start timer T3302.

Aslo, as per 24.008 Sec 4.2.5.17, in the sub state “ATTEMPTING-TO-UPDATE-MM” (When in state GMM-REGISTERED), the UW will initiate a RAU procedure with update type set to “Combined RA/LA updating with IMSI attach” on expiry of timers T3311 or T3302.

Hence, at Step 12, the UE will perform a RAU procedure, NOT an ATTACH procedure. The current TTCN expects an ATTACH, which is incorrect and needs to be changed. A Prose CR will also be raised to adress this issue.

3) In the current TTCN implementation, at Step 8 the ATTACH ACCEPT message is sent with GMM cause congestion (#22). However, the template used for the GMM cause does not include the IEI. This needs to be changes since the GMM Cause is a TV type as per Table 9.4.2 in 24.008 sec 9.4.2. This needs to be changed.

	Summary of change
	1) Modified the TTCN so that Instead of expecting an ATTACH REQ with IMSI and TMSI Status, the ATTACH REQ comes with a PTMSI.
2) Removed the implementation of ATTACH procedure from test Steps 12 to 18 and replaced it with RAU procedure instead.

3) Parameterised the template cs_GMM_Cause so that it can take IEI as an argument. Set the default assignment for IEI to omit so that other usage of this template in other situations is unaffected.

Please see screenshots for details. 

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	function fl_TC_12_2_2_3c_Body() runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var octetstring v_PLMN_A;
    var octetstring v_LAC_A;
    var octetstring v_RAC_A;
    var template (present) RoutingAreaId v_RAI_1;
    var InitialUE_Identity v_InitialUE_Id;
    timer t_T3302Ext_UpperBound := 60.0 * 3.0 * 1.1; // 3 minutes + 10 % tolerance
    timer t_T3302Ext_LowerBound := 60.0 * 3.0 * 0.9; // 3 minutes - 10 % tolerance
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    //@siclog "Step 1" siclog@
    // The UE is set in UE operation mode A
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //@siclog "Step 2" siclog@
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 3" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 4" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach' - Mobile identity = IMSI
    //TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Steps 8-9" siclog@
    // ATTACH ACCEPT:
    //   Attach result = 'GPRS only attached'; Allocated P-TMSI = P-TMSI-1;
    //   P-TMSI Signature = P-TMSI-1 signature; Routing area identity = RAI-1
    //   GMM cause= 'congestion'; T3302= 3min
    // ATTACH COMPLETE
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeGPRSAtt,
                           px_PTMSI_Def,
                           px_PTMSI_SigDef,
                           -,
                           -,
                           cs_GMM_Cause(tsc_GMM_Cause_Congestion),
                           -,
                           cs_GPRS_Timer2_3_IEI('2A'O, tsc_GprsTimerUnit_1min, '00011'B));  //T3302= 3min @sic R5s150329 Baseline Moving sic@
    //Start timers to verify if the UE waits for 3 min before initiating a combined attach procedure.
    t_T3302Ext_LowerBound.start;
    t_T3302Ext_UpperBound.start;
    //@siclog "Step 10" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 11" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
        }
      [] t_T3302Ext_LowerBound.timeout
        {
          alt {
            [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                            tsc_RB0,
                                            cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
              {
                t_T3302Ext_UpperBound.stop;
                v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
                f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
              }
            [] t_T3302Ext_UpperBound.timeout
              {
                //upper bound timer time-out
                f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
              }
          }
        }
    }
    //@siclog "Step 12" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach' - Mobile identity = IMSI
    //TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag == tsc_NoValidTmsi){
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.2.2.3c, Step 12");
    }
    else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 14-15" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 16" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 17" siclog@
    // ATTACH ACCEPT:
    // Attach result = 'Combined GPRS/IMSI attached'
    // Allocated P-TMSI = P-TMSI-1
    // P-TMSI Signature = P-TMSI-1 signature
    // MS identity =IMSI
    // Routing area identity = RAI-1
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeCombined,
                           px_PTMSI_Def,
                           px_PTMSI_SigDef);
  } // fl_TC_12_2_2_3c_Body


After:

	function fl_TC_12_2_2_3c_Body() runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var octetstring v_PLMN_A;
    var octetstring v_LAC_A;
    var octetstring v_RAC_A;
    var template (present) RoutingAreaId v_RAI_1;
    var InitialUE_Identity v_InitialUE_Id;
    var template (value) RRCDataReqType v_RAUAccept;//WA#12_2_2_3c
    var B3_Type v_UpdateType; //WA#12_2_2_3c
    timer t_T3302Ext_UpperBound := 60.0 * 3.0 * 1.1; // 3 minutes + 10 % tolerance
    timer t_T3302Ext_LowerBound := 60.0 * 3.0 * 0.9; // 3 minutes - 10 % tolerance
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    //@siclog "Step 1" siclog@
    // The UE is set in UE operation mode A
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //@siclog "Step 2" siclog@
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 3" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 4" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach' - Mobile identity = IMSI
    //TMSI status = no valid TMSI available
    //WA#12_2_2_3c : Check for PTMSI implemented. Prose CR needed.                                                                                      */

U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          cs_MobileIdTMSIorPTMSI(omit, px_PTMSI_Def)))) -> value v_RRC_DataInd;
    //WA#12_2_2_3c : Enforce a check that PTMSI is being sent. Prose CR needed
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.mobileIdentity)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "P-TMSI not present");      
    }   
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Steps 8-9" siclog@
    // ATTACH ACCEPT:
    //   Attach result = 'GPRS only attached'; Allocated P-TMSI = P-TMSI-1;
    //   P-TMSI Signature = P-TMSI-1 signature; Routing area identity = RAI-1
    //   GMM cause= 'congestion'; T3302= 3min
    // ATTACH COMPLETE
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeGPRSAtt,
                           px_PTMSI_Def,
                           px_PTMSI_SigDef,
                           -,
                           -,
                           cs_GMM_Cause (tsc_GMM_Cause_Congestion, '25'O),//WA#12_2_2_3c
                           -,
                           cs_GPRS_Timer2_3_IEI('2A'O, tsc_GprsTimerUnit_1min, '00011'B));  //T3302= 3min @sic R5s150329 Baseline Moving sic@
    //Start timers to verify if the UE waits for 3 min before initiating a combined attach procedure.
    t_T3302Ext_LowerBound.start;
    t_T3302Ext_UpperBound.start;
    //@siclog "Step 10" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 11" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
        }
      [] t_T3302Ext_LowerBound.timeout
        {
          alt {
            [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                            tsc_RB0,
                                            cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
              {
                t_T3302Ext_UpperBound.stop;
                v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
                f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
              }
            [] t_T3302Ext_UpperBound.timeout
              {
                //upper bound timer time-out
                f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
              }
          }
        }
    }
    //@siclog "Step 12" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach' - Mobile identity = IMSI
    //TMSI status = no valid TMSI available
    //WA#12_2_2_3c : the UE will send a RAU REQ, not an ATACH REQ. Prose CR required.
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, ?), 
                                                                 v_RAI_1,
                                                                 ?))) -> value v_RRC_DataInd;
    v_UpdateType := v_RRC_DataInd.msg.routingAreaUpdateRequest.updateType.updateValue;
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 14-15" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 16" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 17" siclog@
    // ATTACH ACCEPT:
    // Attach result = 'Combined GPRS/IMSI attached'
    // Allocated P-TMSI = P-TMSI-1
    // P-TMSI Signature = P-TMSI-1 signature
    // MS identity =IMSI
    // Routing area identity = RAI-1
     //WA#12_2_2_3c : Send the RAU accept. Prose CR required
     v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResult ('001'B),
                                   v_RAI_1,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def), 
                                   omit);    
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //WA#12_2_2_3c : Handle RAU complete. Prose CR required.
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
  } // fl_TC_12_2_2_3c_Body


4.5 cs_GMM_Cause
	Record Name
	Cs_GMM_Cause

	Reason for change
	In the current TTCN implementation, at Step 8 the ATTACH ACCEPT message is sent with GMM cause congestion (#22). However, the template used for the GMM cause does not include the IEI. This needs to be changes since the GMM Cause is a TV type as per Table 9.4.2 in 24.008 sec 9.4.2. This needs to be changed

	Summary of change
	Parameterised the template cs_GMM_Cause so that it can take IEI as an argument. Set the default assignment for IEI to omit so that other usage of this template in other situations is unaffected.

	Source of change
	NAS_24008Templates_IRAT.ttcn

	
	


Before:
	template (value) GMM_Cause cs_GMM_Cause (B8_Type p_RejectCause) :=
  { // @sic R5s110718 sic@
  /* @status    APPROVED (LTE_IRAT, UTRAN) */
   iei := omit,
   gmmCauseValue := p_RejectCause
  };


After:

	template (value) GMM_Cause cs_GMM_Cause (B8_Type p_RejectCause,
  









   template (omit) IEI8_Type p_IEI := omit) :=
  { // @sic R5s110718 sic@
  /* @status    APPROVED (LTE_IRAT, UTRAN) */
   iei := p_IEI,
   gmmCauseValue := p_RejectCause
  };


5 Branches executed in test case 12.2.2.3c
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel 7360 device
The Intel 7360 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_12_2_2_3c_Intel_7360.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References
	[1]
	R5s160723
This archive comprises Execution log files, PICS/PIXIT files


